DTU Library

The LMT circuit and SPICE

Lindberg, Erik; Murali, K.; Tamacevicius, Arunas

Published in:
Proc. 14th Int. Workshop “Nonlinear Dynamics of Electronic Systems, NDES2006”

Publication date:
2006

Document Version
Publisher's PDF, also known as Version of record

The LMT circuit and SPICE

Erik Lindberg, K. Murali and Arunas Tamacevicius *

Abstract — **The state equations of the *LMT* circuit are modeled as a dedicated analogue computer circuit and solved by means of *PSpice*. The nonlinear part of the system is studied. Problems with the *PSpice* program are presented.**

1 INTRODUCTION

Recently the *LMT* circuit [1] was presented as the smallest transistor-based non-autonomous chaotic circuit. Measurements and *PSpice* simulations were presented. The aim of this workshop-note is to present the state equations for the circuit and transfer these equations into a dedicated analogue computer circuit model for alternative *PSpice* simulations. The simulations demonstrate that the results from a *PSpice* simulation are sensitive with respect to the automatic setup of the equations by *PSpice* e.g. the result depends on the order of the elements in the net-list.



Figure 1: The *LMT* circuit:
An oscillator VS with an active *RC* load composite.
$VS = 10\text{V}/10\text{kHz}$, $R1 = 1\text{k}\Omega$, $C1 = 4.7\text{nF}$,
$R2 = 994\text{k}\Omega$, $C2 = 1.1\text{nF}$, $Q1 = 2N2222A$.
$\tau_0 = (R_1 + R_2)(\frac{C_1 C_2}{C_1+C_2}) = 0.8869224136\text{ms}$.

2 THE STATE EQUATIONS OF THE LMT CIRCUIT

Figure 1 shows the *LMT* circuit [1]. The state equations for the circuit with the capacitor voltages $V(C_1) = V(1,4) = V(1) - V(4) = V_1 - V_4$ and $V(C_2) = V(2,0) = V(2) = V_2$ as variables x_1 and x_2 are found as:

$$\frac{dx_1}{dt} = -\frac{1}{\tau_1} \times x_1 + \frac{1}{\tau_1} \times x_2 \quad (1)$$

$$-\frac{R_2}{\tau_1} \times IC - \frac{1}{\tau_1} \times VS$$

$$\frac{dx_2}{dt} = +\frac{1}{\tau_2} \times x_1 - \frac{1}{\tau_2} \times x_2 \quad (2)$$

$$-\frac{1}{C_2} \times IB - \frac{R_1}{\tau_2} \times IC + \frac{1}{\tau_2} \times VS$$

where $\tau_1 = (R_1+R_2)\times C_1$ and $\tau_2 = (R_1+R_2)\times C_2$. IC and IB are the currents into the transistor collector and base terminals respectively. If the $Ebers - Moll$ injection model is used for the transistor, IC and IB are nonlinear functions of the node voltages V_1 and V_2 and they are calculated from the following equations:

$$IC = -IBC + \alpha_f \times IBE \quad (3)$$

$$IB = (1 - \alpha_f) \times IBE + (1 - \alpha_r) \times IBC \quad (4)$$

$$IBE = I_s \times \left(exp\left(\frac{V_2}{VT}\right) - 1 \right) \quad (5)$$

$$IBC = I_s \times \left(exp\left(\frac{V_2 - V_1}{VT}\right) - 1 \right) \quad (6)$$

The node voltage $V(1)$ is introduced as variable x_3:

$$x_3 = -R_2 \times C_1 \times \frac{dx_1}{dt} + x_2 - R_2 \times IC \quad (7)$$

The equations my be rearranged as follows:

$$\left\{ \begin{array}{c} \frac{dx_1}{dt} \\ \frac{dx_2}{dt} \end{array} \right\} = \left\{ \begin{array}{cc} -\frac{1}{\tau_1} & +\frac{1}{\tau_1} \\ +\frac{1}{\tau_2} & -\frac{1}{\tau_2} \end{array} \right\} \left\{ \begin{array}{c} x_1 \\ x_2 \end{array} \right\} \quad (8)$$

$$+\left\{ \begin{array}{c} F_1(x_1, x_2) \\ F_2(x_1, x_2) \end{array} \right\} + \left\{ \begin{array}{c} -\frac{1}{\tau_1} \times VS \\ +\frac{1}{\tau_1} \times VS \end{array} \right\}$$

where $\tau_1 = (R_1+R_2)\times C_1$ and $\tau_2 = (R_1+R_2)\times C_2$. $F_1 = -\frac{R_2}{\tau_1} \times IC$ and $F_2 = -\frac{1}{C_2} \times IB - \frac{R_1}{\tau_2} \times IC$ are nonlinear functions of x_1 and x_2. The eigenvalues of the linear part becomes: $\lambda_1 = 0$ and $\lambda_2 = \frac{1}{(R_1+R_2)(\frac{C_1 C_2}{C_1+C_2})}$ as expected.

*Ørsted•DTU Department, 348 Technical University of Denmark, DK2800 Kongens Lyngby, DENMARK, e-mail: el@oersted.dtu.dk, tel.: +45 4525 3650, fax: +45 4588 0117.



Figure 2: The *LMT* circuit: *IC* as function of *VCE*. Analysis based on differential equations. Oscillator *VS* placed at $*-1$ in the netlist below.

3 PSPICE INPUT FILES

The equations above are transformed into a dedicated ideal analogue computer model as shown in the following input file for *PSpice*, the netlist.

```
LMT circuit, SPICE model based on
*            differential equations
*
  .tran     0      83m    20m   1u
* .tran     0      20m      0   1u
*             1u < 100u < 886.9u
.probe
.options nopage opts
+ RELTOL =   1.00000E-05    ;  03
+ ABSTOL =    1.0000E-12
+  VNTOL =    1.0000E-12    ;  06
+   GMIN =    1.0000E-12  ITL4 = 1000
+ CHGTOL =   10.0000E-15
+ NUMDGT = 4  ITL1 = 1500  ITL2 =   200
* -------------------------------------
* ------------------------ define source
*
*-1 VOSC 20 0 sin ( 0  10  10e+3  0  0 )
*                          T = 100us
* ---------- define variables, integrate
*               V3 = X1 = V(C1) = (V1-V4)
 CX1  3  0  1     ;
 RCX1 3  0  1e+20 ; SPICE resistor ;-)
 CX2  2  0  1     ; V2 = X2 = V2
 RCX2 2  0  1e+20 ; SPICE resistor ;-)
 RX3  1  0  1     ; V1 = X3 = V1
* -------------------------------------
* - calculate nonlinear part IBE and IBC
*
.model diode d  ( RS=1 IS=14.34e-15 )
*
 EV2   4  0    2  0   1
 VIBE  4  6    dc 0   ; measure IBE
 DBE   6  0    diode
 EV21  5  0    2  1   1
 VIBC  5  7    dc 0   ; measure IBC
 DBC   7  0    diode
*
* ----------------  calculate IB and IC
*
 RIB  8  0  1     ;  V(RIB) = V(8) = IB
 FBE  0  8  VIBE  3.892565e-3   ;  1-AF
 FBC  0  8  VIBC  0.1410039480  ;  1-AR
 RIC  9  0  1     ;  V(RIC) = V(9) = IC
 FCC  0  9  VIBC  -1
 FCE  0  9  VIBE  +0.9961074348  ;  +AF
*
*-2 VOSC 20 0 sin ( 0  10  10e+3  0  0 )
*
* --------------------  calculate dx1/dt
*  tau1 = (R1+R2)C1,
*  R1=1e+3,   R2=994e+3,
*  C1=4.7e-9, C2=1.1e-9
*
 RX1D 10  0      1       ; V(10) = X1DOT
 GOSC1 0 10 20 0 -213.8351331  ; -1/tau1
 GIC1 0 10  9 0 -212.5521223e+6;-R2/tau1
 GX21 0 10  2 0 +213.8351331   ; +1/tau1
 GX11 0 10  3 0 -213.8351331   ; -1/tau1
*
* --------------------  calculate dx2/dt
*  tau2 = (R1+R2)C2,
*  R1=1e+3,   R2=994e+3,
*  C1=4.7e-9, C2=1.1e-9
*
 RX2D 11  0      1       ; V(11) = X2DOT
 GOSC2 0 11  20 0 +913.6592051 ; +1/tau2
 GIC2 0 11  9 0 -913.6592051e+3;-R1/tau2
 GIB2 0 11  8 0 -909.0909091e+6; -1/C2
 GX22 0 11  2 0 -913.6592051   ; -1/tau2
 GX12 0 11  3 0 +913.6592051   ; +1/tau2
*
* calculate X3 = V1
*           V(R3) = V(12) = X3 = V1
*
 R3  12  0            1      ;   V1
 GIC  0 12   9 0   -994e+3   ;  -R2
 GX2  0 12   2 0    +1        ;  +1
 GX1D 0 12  10 0   -4.6718e-3 ;  -(R2*C1)
*
* feed-back
*
  GIX1     0   3    10  0    1
  GIX2     0   2    11  0    1
  GIX3     0   1    12  0    1
* ---------------------------------------
*-3 VOSC 20 0 sin ( 0  10  10e+3  0  0 )
*
.end
```



Figure 3: The *LMT* circuit: *IC* as function of *VCE*. Analysis based on differential equations. Oscillator *VS* placed at $*-2$ in the netlist above.

The Figures 2, 3 and 4 show the result of an experiment with *PSpice*. The input source $VS = VOSC$ is specified in 3 different places in the netlist. It is seen that the result depends on the placement of VS in the netlist.



Figure 4: The *LMT* circuit: *IC* as function of *VCE*. Analysis based on differential equations. Oscillator *VS* placed at $*-3$ in the netlist above.

The following input file for *PSpice* shows an experiment with the specification of the time analysis. The final time is 30ms. The maximum integration step is 0.1μs.

```
LMT model based on circuit elements
*
*-1  .tran   0  30m   0      0.1u
*-2  .tran   0  30m   15m    0.1u
*
.probe
.options nopage opts
+  RELTOL =   1.00000E-05    ;  03
+  ABSTOL =    1.0000E-12
+   VNTOL =    1.0000E-12    ;  06
+    GMIN =    1.0000E-12  ITL4 = 1000
+  CHGTOL =   10.0000E-15
+  NUMDGT = 4  ITL1 = 1500  ITL2 = 200
* -----------------------------------
* define source
*
  VS  5  0  sin ( 0  10  10e+3  0  0 )
*
  R1   5  4        1e+3    ;
  C1   4  1        4.7e-9  ;
  R2   1  2        994e+3  ;
  C2   2  0        1.1e-9  ;
*
  VIC  1  8  dc    0
  VIB  2  6  dc    0
  VIBC 7  8  dc    0
  VIBE 9  0  dc    0
*
.model diode d  ( RS=1 IS=14.34e-15 )
*
  DBC  6  7  diode
  DBE  6  9  diode
  FCB  8  6  VIBE  0.9961074348  ;  AF
  FEB  0  6  VIBC  0.8589960520  ;  AR
* -----------------------------------
* calculation of nonlinear functions
*           F1=-(R2/tau1)*IC and
*           F2=-(1/C2)*IB-(R1/tau1)*IC
 RF1 21 0 1
 F1  0 21 VIC -212.5521223e+6;-R2/tau1
*
 RF2 22 0 1
 F2A 0 22 VIB -909.0909091e+6; -1/C2
 F2B 0 22 VIC -913.6592051e+3;-R1/tau2
*
.end
```



Figure 5: The *LMT* circuit: V2 as function of time Analysis based on circuit elements.
*-1; .tran 0 30m 0 0.1u

In the first analysis $(*-1)$ the whole calculated table is transferred to the probe program. In the second analysis $(*-2)$ only the last part of the table from 15ms to 30ms is transferred. According to the users manual the two analyses should give rise to the same table but the figures 5 and 6 show that this is not the case.

Figure 7 shows the nonlinear functions $F1$ and $F2$ as functions of time. It is seen that extremely



Figure 6: The LMT circuit: V2 as function of time Analysis based on circuit elements.
*-2; .tran 0 30m 15m 0.1u



Figure 7: The LMT circuit: The nonlinear functions F1 = -(R2/tau1)*IC = V(21) and F2 = -(1/C2)*IB-(R1/tau1)*IC = V(22) as function of time.



Figure 8: The LMT circuit: **Limit cycle**
VS 5 0 sin (0 10 **2.1014e+3** 0 0)
.tran 0 200m 20m 10u, $RELTOL$ = 1.00000E-**05**



Figure 9: The LMT circuit: **Chaos**
VS 5 0 sin (0 10 **2.1014e+3** 0 0)
.tran 0 200m 20m 10u, $RELTOL\ =\ 1.00000E{-}\mathbf{06}$

large spikes of about 500k occurs making the solution very sensitive due to the rapid variation of the derivatives $\frac{dx_1}{dt}$ and $\frac{dx_2}{dt}$.

In [1] it is stated that it is difficult to find limit cycle behavior by means of simulation. The figures 8 and 9 show an experiment where a limit cycle is found. It is interesting to observe that it is extremely dependent on the relative accuracy. The limit cycle is found for $RELTOL = 1e-5$. The curve seems to be a little "noisy" so $RELTOL$ is lowered to $1e-6$ in order to obtain a more accurate result. Surprisingly (?) the result is chaos. With $.tran\ \ 0\ \ 200m\ \ 150m\ \ 1u$, $VS\ \ 5\ \ 0\ \ sin$ (0 10 **2.1310e+3** 0 0) and $RELTOL = 1e-6$ a "better" limit cycle is found but it is still "noisy, chaotic" when you zoom i.e. the frequency must be specified with more significant digits before you can distinguish between "physical chaos" and "numerical noise".

4 CONCLUSIONS

The state equations of the LMT circuit are presented and transferred into a dedicated ideal analogue computer circuit for alternative $PSpice$ simulations. The nonlinear part of the equations is studied indicating large sensitivity for the derivatives due to large spikes. Problems with $PSpice$ have been detected. The results seem to be surprisingly sensitive to the order of placement of the input file lines and to the relative tolerance $RELTOL$. Limit cycle behavior is studied. It is very difficult to find proper limit cycles by simulation.

References

[1] Erik Lindberg, K. Murali and Arunas Tamasevicius, "The Smallest Transistor-Based Nonautonomous Chaotic Circuit", IEEE Transactions on Circuits and Systems - II: Express Briefs, vol. 52, no. 10, pp. 661–664, October 2005.